Tutorial 6: Branch Instructions

New Instructions Introduced 

B

BC

New Test Conditions Introduced

BEQ

ANEQ
BNEQ

AGT
BGT

ALT
BLT

ALEQ
BLEQ

AGEQ BGEQ

Overview of Tutorial

In the course of a program there is often the need for decision making, the outcome of which controls the flow of the program. In a high level language, decision making can be written in terms of an IF - THEN - ELSE type of structure.

To implement decision making using TMS320C5000 assembly language, we employ two types of branch instructions: conditional branches and unconditional branches. This tutorial shows how to use both types of branch instructions.

Unconditional  Branches

Readers may well be familiar with the concept of the goto statement in BASIC. When a goto is encountered, rather than falling through to the next line, the program execution is redirected to the program memory address specified as part of the instruction goto. 

On the TMS320C5000, the instruction B (branch unconditionally) is the assembly language equivalent of the goto in BASIC. When an instruction B is encountered, the program execution continues at the address specified by the operand. An example of how to use the instruction B is shown in Example 6-1:

Example 6-1.

	
	
	

	
	LD #100h, A
	; Load accumulator A with 100h.

	
	B end
	; Branch to the label end.

	
	STM #20, AR2
	; This instruction is skipped.

	end:
	LD #30h, B
	; This instruction is the one 

; executed immediately after the 

; instruction B (branch

; unconditionally).


The instruction B takes a single operand - the address to which program execution is redirected, which is specified by a label. In Example 6-1, when the instruction B end is encountered, the next instruction to be executed is the one at the label end. All instructions between the instruction B and the associated label end are skipped (not executed).

The label to which the program branches must appear in the first column of the page. A special symbol $ may be used as the label and indicates that the branch is to the same address as the instruction B; in other words, the instruction B $ branches to itself. The two unconditional branches in Example 6-2 are therefore equivalent:

Example 6-2.

	
	
	

	
	B $
	; Branch to current address.

	done:
	B done
	; Branch to the label done.


In either case, the branch will be to the same address as the instruction itself, so we have infinite loops. Branching to the same address may be used at the end of a program or to lock-up the program when a fatal error occurs.

The label $ can also be used in the expression B $+2 (branch to the address at the program counter + 2) or B $-2(branch to the address at program counter - 2). 

Conditional Branches

We may wish for a branch to occur under certain circumstances but not under others. Whether we branch depends upon the value of a certain variable. This implies that some form of test is built into the process.

To implement a branch that only occurs when a certain condition is met, we use the instruction BC (branch conditionally). This takes two operands, and a typical usage is shown in Example 6-3:

Example 6-3.

	
	

	BC label, condition
	; If condition is true then

; branch to label, otherwise drop

; through to the next instruction. 


The first operand is the address (as denoted by a label). When a branch occurs, this is where the next instruction will be executed. It works in the same way as the label used with the instruction B.

The second operand is the test and this determines whether a branch occurs. On the TMS320C5000, the most convenient variables to test are accumulator A and accumulator B. Tests for other parameters e.g. the carry (C) flag and overflow flags (OVA or OVB) are possible and will be used in later tutorials. For the purposes of this tutorial, we shall restrict all tests to accumulator A and accumulator B. 

Note that we would never actually type in the word condition, but would in fact type in one of a range of special operands. An example of these special operands is AEQ. This will evaluate to TRUE when the contents of accumulator A are zero. For all values of accumulator A except zero, the condition AEQ will evaluate to FALSE. 
Testing for Equality 

Let us start with a simple example of an if statement. If the value of variable x is 5, then increment x. In C code we could write this as:

Example 6-4.

	
	

	unsigned int x;
	 // Allocate a variable x.

	if ( x == 5) 
	 // Test if the value of x is 5.

	  { x++; }
	 // If so, increment variable x.


Depending upon the value of the variable x we carry out an action. 

To implement Example 6-4 in assembly language, we must first assign a data memory location to store the variable x. Let us use data memory address 130h. In order to test the value of the variable x, we must copy it from data memory to either accumulator A or accumulator B. An implementation is given in Example 6-5: 

Example 6-5.

	
	
	

	
	LD #2, DP
	; Page 2. Gain access to data memory

; addresses 100h to 17Fh.

	
	LD 30h, A
	; Copy variable x, stored at data

; memory address 100h + 30h =

; 130h into accumulator A.

	
	SUB #5, A
	; Subtract 5 from variable x.

	
	BC incr, AEQ 
	; If the value in accumulator A is

; equal to zero (the condition AEQ

; is TRUE), then branch to the label

; incr. Otherwise, drop through to

; the next line.

	
	B done
	; No action to be taken. Skip next

; instruction.

	incr:
	ADDM #1, 30h
	; Increment variable x.

	done:
	
	; Processing complete. 


In Example 6-5, the instruction B (branch conditionally) uses the label incr as the first operand. When a branch occurs, it will be to this label. The second operand is the particular test, in this case the condition AEQ (accumulator A is equal to zero).   

If the condition AEQ evaluates to TRUE, then program execution continues at the program address specified by the label incr. On the other hand, if the test condition does not evaluate to TRUE, the branch does not occur and program execution continues at the next line.    

We can in fact write the code in Example 6-5 in a slightly neater way that saves one instruction. Let us this time use accumulator B for the test. The operand we shall use with the instruction BC (branch conditional) is BNEQ (accumulator B not equal to zero). 

Example 6-6.

	
	
	

	
	LD #2, DP
	; Page 2. Gain access to data

; memory addresses 100h to 17Fh.

	
	LD 30h, B
	; Copy variable x at data memory

; address 100h + 30h = 130h into

; accumulator B.

	
	SUB #5, B
	; Subtract 5 from the copy of

; variable x in accumulator B.

	
	BC done, BNEQ 
	; If the value in accumulator B is

; not equal to zero (the condition

; BNEQ is TRUE), then branch to the

; label done. Otherwise, drop

; through to the next line.

	
	ADDM #1, 30h
	; Increment variable x.

	done:
	
	; Processing complete. 


In the case of Example 6-6, we branch on the opposite condition to which we wish to carry out the operation. We have used the test for the condition BNEQ (accumulator B not equal to zero) to go straight to the label done if the value of the variable is not 5. This has eliminated the instruction B (branch unconditionally). Branching on the opposite condition is often used by the TMS320C5000 C compiler to reduce the amount of code required.

In Examples 6-4, 6-5 and 6-6, we have performed a comparison of the variable x with 5 using the instruction SUB (subtract from accumulator).  Another way to perform the comparison is to use the instruction XOR (logical exclusive OR accumulator). In this case, we could have written Example 6-6 as: 

Example 6-7.

	
	
	

	
	LD #2, DP
	; Page 2. Gain access to data

; memory addresses 100h to

; 17Fh.

	
	LD 30h, B
	; Copy variable x at data memory

; address 100h + 30h = 130h into

; accumulator B.

	
	XOR #5, B
	; Compare 5 with the copy of

; variable x.

	
	BC done, BNEQ 
	; If the value in accumulator B is

; not equal to zero (the condition

; BNEQ is TRUE), then branch to the

; label done. Otherwise, drop

; through to the next line.

	
	ADDM #1, 30h
	; Increment variable x.

	done:
	
	; Processing complete. 


Care does need to be taken using the instruction XOR (exclusive OR) for comparisons. The instruction XOR tests only the low word of the accumulator. Therefore, if the value to be tested in the accumulator is greater than 0000FFFFh, the instruction SUB (subtract from accumulator) should be used instead.

Testing for Inequality

In Example 6-4, we carried out an action if variable x was a certain value. We can also test for the opposite condition, as shown in Example 6-8:  

Example 6-8.

	
	

	unsigned int x;
	 // Allocate a variable x.

	if ( x != 5) 
	 // If the value of x is not 5 then

	  { x++; }
	 // increment variable x.


The code to implement this is similar to that of Example 6-6, except that in this case we use the operand BEQ (accumulator B equal to zero). We shall also use indirect addressing:  

Example 6-9.

	
	
	

	
	STM #130h, AR5
	; Store address in data memory

; of variable x in auxiliary

; register AR5. AR5 = 130h. 

	
	LD *AR5, B
	; Copy variable x into accumulator

; B.

	
	SUB #5, B
	; Subtract 5 from copy of variable

; x.

	
	BC done, BEQ 
	; If the value in accumulator B is

; equal to zero (the condition BEQ

; is TRUE), then branch to the

; label done. Otherwise, drop

; through to the next line.

	
	ADDM #1, *AR5
	; Increment variable x.

	done:
	
	; Processing complete 


Using an IF-ELSE Structure

Let us now make the program flow slightly more complex. If the value of variable y is  equal to 20, then increment y. Otherwise, decrement y. In C code we would write this as shown in Example 6-10:

Example 6-10.

	
	

	unsigned int y;
	// Declare a variable.

	if ( y == 20 ) 
	// Test if x is equal to 20.

	  { y++ ; }
	// Increment variable y.

	else 
	

	  { y--; }
	// Decrement variable y.


Depending upon the value of the variable y we carry out an action. 

To implement Example 6-10 in assembly language, to test the value of the variable y, we must first copy it to the accumulator. If the variable y is stored at data memory address 2FFh, then the assembly language implementation of Example 6-10 is shown in Example 6-11: 

Example 6-11.

	
	
	

	
	LD #5, DP
	; Page 5. Access to data memory

; addresses 280h to 2FFh.

	
	LD 7Fh, A
	; Copy variable y at data memory

; address 280h + 7Fh = 2FFh into

; accumulator A.

	
	XOR #20, A
	; Compare variable y with 20. If

; equal, accumulator A will then

; contain zero.

	
	BC next, ANEQ 
	; If the value in accumulator  A

; is not equal to zero (the

; condition ANEQ is TRUE), then

; branch to the label next.

; Otherwise, drop through to the

; next line.

	
	ADDM #1, 7Fh
	; Increment variable y.

	
	B end
	; No more processing to do.

	
	SSBX SXM
	; Turn on sign-extension mode.

	next:
	ADDM #0FFFFh, 7Fh
	; Subtract 1 from y. 

; y = y + (-1).  

	done:
	
	; End of processing.


Note that in Example 6-11 we have used the instruction ADDM (add long immediate value to memory) to subtract 1 from the variable y. There is no such instruction as SUBM (subtract long immediate from memory). Therefore we have turn on sign-extension mode to add -1 (FFFFh) to the contents of the data memory address in order to decrement variable y.     

Testing for Greater Than

So far we have tested for equality and inequality when branching. There are cases where we need to test for other conditions such as greater than or less than. 

Let us take a practical example. We have an input that we wish to give a maximum limit to avoid overloads. We can limit the input as shown in Example 6-12:

Example 6-12.

	
	

	unsigned int input;
	// Declare a variable.

	if ( input > 1000 )
	// Range check input.

	 { input = 1000; }
	// Truncate if too great.


Assuming our variable input is stored in data memory address 310h, we can implement the C code in Example 6-12 as shown in Example 6-13:

Example 6-13.

	
	
	

	
	LD #6, DP
	; Page 6. Access to data memory

; addresses 300h to 37Fh. 

	
	LD 10h, A
	; Copy variable input at data

; memory address 300h + 10h = 310h

; into accumulator A.

	
	SUB #1000, A
	; Accumulator A - 1000.

	
	BC limit, AGT
	; If input is greater than 1000

; then branch to label limit. 

	
	B done
	; No processing if variable input
; is not greater than 1000.

	limit:
	ST #1000, 10h
	; Limit value in variable input to

; a maximum of 1000.

	done: 
	
	; No more processing to do.


The operand used with the instruction BC that causes a branch with opposite sense to that of operand AGT (accumulator A greater than zero) is ALEQ (accumulator A is less than or equal to zero). Let us repeat the operation as shown in Example 6-13, but this time using indirect addressing with accumulator B. In order to eliminate the instruction B (branch unconditional) to reduce the code size, we shall use the instruction BC (branch conditional) with the operand BLEQ  (accumulator B is less than or equal to zero):

Example 6-14.

	
	
	

	
	STM #310h, AR2
	; Store address of variable input ; in auxiliary register AR2. AR2 =

; 310h. 

	
	LD *AR2,  B
	; Copy variable input into

; accumulator B.

	
	SUB #1000, B
	; Accumulator B - 1000.

	
	BC done, BLEQ
	; If input is less than or equal

; to 1000 (the condition BLEQ  is

; TRUE) then branch to the label

; done. 

	
	ST #1000, *AR2
	; Limit value in variable input to

; 1000.

	done: 
	
	; No more processing to do.


Testing for Less Than

Let us take another example. Consider the case where we have a variable to which we wish to apply a lower limit of 1000.  

Example 6-15.

	
	

	unsigned int input1;
	// Declare a variable.

	if ( input1 < 1000 )
	// Range check input.

	 { input1 = 1000; }
	// Apply lower limit if 

// too small.


Assuming our variable input1 is stored at data memory address 181h, we can implement the C code in Example 6-15 as:

Example 6-16.

	
	
	

	
	LD #3, DP
	; Page 3. Access to data memory

; addresses 180h to 1FFh.

	
	LD 1h,  A
	; Copy variable input1 at data

; memory address 180h + 1h = 181h

; into accumulator A.

	
	SUB #1000, A
	; Accumulator A - 1000.

	
	BC limit, ALT
	; If input1 is less than 1000 (the

; condition ALT is TRUE) then

; branch to the label limit. 

	
	B done
	; No processing if variable input1
; is greater than or equal to

; 1000.

	limit:
	ST #1000, 1h
	; Limit value in variable input1 to

; 1000.

	done: 
	
	; No more processing to do.


The operand causing a branch with the opposite sense to the operand ALT (accumulator A less than zero) is AGEQ (accumulator A greater or equal to zero). 

Let us repeat the operation in Example 6-16, this time using indirect addressing with accumulator B and the operand BGEQ (accumulator B greater or equal to zero):

Example 6-17.

	
	
	

	
	STM #181h, AR5
	; Store address of variable input1

; in auxiliary register AR5. AR5 =

; 181h.

	
	LD *AR5, B
	; Copy variable input1 into

; accumulator B.

	
	SUB #1000, B
	; Accumulator B - 1000.

	
	BC done, BEQ
	; If input1 is greater than or

; equal to 1000 (the condition BEQ

; is TRUE) then branch to the

; label done. 

	
	ST #1000, *AR5
	; Apply lower limit of 1000 to

; variable input1.

	done: 
	
	; No more processing to do.


Testing for Greater or Equal 

This time we shall test the condition where the variable under test is greater than or equal to a certain value, and if so, take action accordingly. Consider the case of a water heating system where if the water temperature is 95 Centigrade degrees or more, we require the heater output to turn down to half power (50%). This is shown in the C code in Example 6-18: 

Example 6-18.

	
	

	unsigned int temperature;
	// Input variable.

	unsigned int heater;
	// Output variable.

	if ( temperature >= 95 )
	// Range check input.

	 { heater = 50; }
	// Set output accordingly.


Assuming our variable temperature  is stored in data memory address 144h and the variable heater  at 145h, we can implement Example 6-18 as:

Example 6-19.

	
	
	

	
	LD #2 ,DP
	; Page 2. Gain access to data

; memory addresses 100h to

; 17Fh.

	
	LD 44h,  A
	; Copy variable temperature at

; data memory address 100h + 44h =

; 144h into accumulator A.

	
	SUB #95, A
	; Accumulator A - 95.

	
	BC act, AGEQ
	; If temperature is greater than

; 95 (the condition AGEQ is TRUE)

; then branch to the label act. 

	
	B done
	; No processing if temperature
; is less than 95.

	act:
	ST #50, 45h
	; Set value in variable

; heater to 50.

	done: 
	
	; No more processing to do.


The operand that causes a branch with the opposite sense to the operand AGEQ (accumulator A greater or equal to zero) is ALT (accumulator A less than zero). By using the instruction BC (branch conditional) with the operand of opposite sense we can reduce the size of the code by doing away with the instruction B. This time we shall use indirect addressing with accumulator B, so we need the corresponding operand BLT (accumulator B less than zero)

Example 6-20.

	
	
	

	
	STM #144h, AR0
	; Store address of variable

; temperature in auxiliary

; register AR0. AR0 = 144h.

	
	LD *AR0+,  B
	; Copy variable temperature into

; accumulator B. Increment AR5 to

; contain the address of variable

; heater. AR5 = 145h. 

	
	SUB #95, B
	; Accumulator B - 95.

	
	BC done, BLT
	; If temperature is less than 95

; (the condition BLT is TRUE) then

; branch to the label done. 

	
	ST #50, *AR0
	; Set value in variable heater to

; 50.

	done: 
	
	; No more processing to do.


Testing for Less than or Equal to 

Finally, we shall test the condition where the variable under test is less than or equal to a certain value, and if so, take action accordingly. A practical application might be a frost alarm, where if the air temperature is 5 degrees Centigrade or lower then we turn the heater on to full (100%). The C code to implement this is shown in Example 6-21.

Example 6-21.

	
	

	unsigned int temperature;
	// Input variable.

	unsigned int heater;
	// Output variable.

	if ( temperature <= 5 )
	// Range check temperature.

	 { heater = 100; }
	// Set heater to 100 percent.


Assuming our variable temperature is stored in data memory location 144h and heater at 145h, we can implement the C code in Example 6-21 as:

Example 6-22.

	
	
	

	
	LD #2, DP
	; Page 2. Gain access to data

; memory addresses 100h to 17Fh.

	
	LD 44h,  A
	; Copy variable temperature at

; data memory address 100 + 44h =

; 144h into accumulator A.

	
	SUB #5, A
	; Accumulator A - 5.

	
	BC frost, ALEQ
	; If temperature is less than or

; equal to 5 degrees then branch

; to the label frost. 

	
	B done
	; No processing if variable

; temperature is greater than 5.

	frost:
	ST #100, 45h
	; Set value in variable heater to

; 100 percent .

	done: 
	
	; No more processing to do.


The operand with the opposite sense to the operand ALEQ (accumulator A less than or equal to zero) is AGT (accumulator A greater than zero). By using the instruction BC (branch conditionally) with the operand of opposite sense we can reduce the code by getting rid of the instruction B done. This time we shall use indirect addressing with accumulator B, so we use the operand BGT (accumulator B greater than zero), as illustrated in Example 6-23. 

Example 6-23.

	
	
	

	
	STM #144h, AR1
	; Store address in data memory

; of variable temperature
; in auxiliary register AR1.

; AR1 = 144h.

	
	LD *AR1+, B
	; Copy variable temperature into

; accumulator B. Increment AR1 to

; point to the variable heater.

	
	SUB #5, B
	; Accumulator B - 5.

	
	BC BGT, done
	; If variable temperature is

; greater than to 5 degrees

; then branch to the label done. 

	
	ST #100, *AR1
	; Set value in variable heater to

; 100.

	done: 
	
	; No more processing to do.


Upgrading from the TMS320C2000 to the TMS320C5000

The syntax of the conditional branch instructions does vary across the devices. A table of the equivalent branch instructions is given in Table 6-1:

Table 6-1. Comparison of Instructions

	Description
	TMS320C2000 Instruction 
	TMS320C5000 

Instruction

	Branch unconditional
	B
	B

	Branch if accumulator is equal to zero
	BCND EQ
	BC AEQ  or  

BC BEQ

	Branch if accumulator is 

not equal to zero
	BCND NE
	BC ANEQ or 

BC BNEQ

	Branch if accumulator is 

less than zero
	BCND LT
	BC ALT or 

BC ALT

	Branch if accumulator is 

less than or equal to zero
	BCND LE
	BC ALEQ or 

BC BLEQ

	Branch if accumulator is greater than zero
	BCND GT
	BC AGT or 

 BGT

	Branch if accumulator is greater than or equal to zero
	BNCD GE
	BC AGEQ 

or BC BGEQ 


Because the TMS320C5000 has two accumulators, there are two sets of conditions for each branch instruction. 

Questions  

	1. 
	What is the difference between a conditional branch and an unconditional branch?

	2.
	What is meant by the symbol $?

	3.
	The instruction BC loop4,AEQ (branch accumulator A equal) means equal to what?

	4.
	Why do we use the instruction SUB (subtract from accumulator) rather than the instruction XOR (exclusive OR with accumulator) in order to test the value in an accumulator? 

	5.
	How would we test if the value stored at data memory address 75h is equal to 21?

	6.
	How would we test if the value stored at data memory address 76h is not equal to 5? 

	7.
	Write code to test whether the contents of data memory address 77h are greater than 29, and if so, load accumulator A with 0. 

	8.
	Write code to test if the contents of data memory address 78h are less than 300h, and if so, load accumulator B with 33h. 

	9.
	How would we test if the value stored at data memory address 79h is greater than or equal to 1000h?

	10.
	How would we test if the value stored at data memory address 80h is less than or equal to 1Fh?

	11.
	Using accumulator A, what is the opposite condition to:

a) 
ANEQ
b) 
AEQ
c) 
AGT
d) 
ALT
e) 
AGEQ
f) 
ALEQ

	12.
	In several of the examples in this tutorial we have chosen to branch on the 

opposite condition to the one required to carry out an action. Why should this be the case?

